

# The accurate Electro-Thermal Model of Merged SiC PiN Schottky Diodes

M. Zubert, L. Starzak, G. Jablonski, M. Napieralska, M. Janicki and A. Napieralski

Department of Microelectronics and Computer Science, Technical University of Lodz  
ul. Wolczanska 221/223, 90-924 Lodz, Poland, [mariuszz@dmcs.p.lodz.pl](mailto:mariuszz@dmcs.p.lodz.pl)

## ABSTRACT

The paper presents a novel SPICE model for SiC merged PiN Schottky diodes dedicated to the dynamic as-well-as very accurate static simulation. The model takes into account the temperature dependence of device characteristics and combines in a single model the behavior typical for bipolar and unipolar devices. The presented dynamic simulations of the diode switching process demonstrate that the proposed model produces accurate simulations results, consistent with the measurements. On the contrary, the model provided by the manufacturer fails to predict properly the device behavior.

**Keywords:** SiC, PiN, MPS diode, Schottky Diodes, Static Electro-Thermal Model, SPICE

## 1 INTRODUCTION

Silicon carbide devices are the most promising semiconductor devices for power industrial applications. These devices offer, at least theoretically, excellent thermal properties and high operating frequencies as-well-as high power levels. The most frequently used SiC devices are the Merged PiN Schottky (MPS) diodes. Unfortunately, still there are no available models, which would be able to accuracy predict device behaviour in a relatively wide range

### Electric domain:




Figure 1: Simplified model topology (only steady-state dependences).

of operating temperatures. On the other hand, the correct prediction of the device behaviour is required for the robust design of state-of-the-art power equipment. Obviously, there are models provided by manufacturers and the classical SPICE embedded diode models as-well-as physical models, but none of them is able to produce accurate temperature dependent device characteristics for SiC MPS.

## 2 DYNAMIC MODEL

The very accurate steady-state electro-thermal behavioral model (2-nd and 3-rd generation), measurement procedure as-well-as parameter extraction have been presented in [1] and [2] respectively. The proposed static model of SiC MPS diodes (Fig. 1) showed a very good agreement of the simulated forward and reverse characteristics with the measurements of real devices (Fig. 2) and definitely produced much better results than the models provided by the device manufacturers (Fig. 3) as-well-as physical models. The dynamic behavior has been obtained by the junction charging model (for more see listing in Table II; device: cqj)

$$C_j = C_{j0} \cdot |V(511) + VJ|^{1-MJ} \quad (1)$$

	<b>SDP04S60</b>	<b>SDP10S30</b>	<b>CSD04060</b>	<b>CSD10030</b>	<b>C3D04060</b>
<b>R<sub>s</sub></b>	0.02533977	0.02656187	0.02464326	0.02529571	0.0253904
<b>R<sub>Th</sub></b>	4.112	2.3	2.4	1.9	2.02
<b>Reverse bias:</b>					
<b><math>\alpha_2</math></b>	5.41723813568926E-08	2.10144959397976E-07	see [3]	2.09920550874107E-07	0
<b><math>\alpha_1</math></b>	-7.80429621880507E-5	-1.72626905326908E-4		-1.88345913175174E-4	-4.127549E-05
<b><math>\alpha_0</math></b>	0.0291607049966412	0.0633278592970076		0.0322715235840343	1.778526E-02
<b><math>\beta_1</math></b>	2.57980343601669E-15	1.80333526117792E-14		6.29080916898752E-11	1.521115E-12
<b><math>\beta_2</math></b>	0.064468696717636	0.0564502636696925		0.08215764156411	3.894504E-02
Comment	25±150°C; CL 0.99	25±150°C; CL 0.99	25±150°C	25±150°C; CL 0.99	25±150°C
<b>Forward bias:</b>					
<b>V<sub>intrsc1</sub></b>	1.02334	1.069501	0.9699216	0.6585694	0.948265
<b>V<sub>intrsc2</sub></b>	-1.287796E-3	-1.264594E-3	-1.118851E-3	-1.109617E-3	-1.042935E-3
<b>V<sub>ref</sub></b>	0.2525095	0.1293532	0.2130102	0.1399051	0.0273944
<b><math>\alpha_{ref,1}</math></b>	2.572823E-3	1.154090E-3	2.688235E-3	2.842473E-3	2.70945E-3
<b><math>\alpha_{ref,2}</math></b>	1.849108E-5	-7.411367E-7	4.285593E-5	-4.556804E-6	0
Comment	Measurements 5098; -2÷125°C; CL 0.99	Measurements 420; 25±150°C; CL 0.95	Measurements 235; 25±150°C; CL 0.99	Measurements 366; 25±150°C;CL 0.99	Measurements 1825; 25±150°C;CL 0.99
Dynamic behaviour (based on datasheets, VJ estimated for T=300K) Qj=Cj0*PWR(VJ,MJ)*PWR(V(511)+VJ,1-MJ)/(1-MJ),					
<b>cj0</b>	150.0E-12	600.0E-12	220.0E-12	660.0E-12	251E-12
<b>vj</b>	0.6352922	0.524021	0.623947	0.216542	0.375303
<b>mj</b>	0.442673	0.423875	0.370342	0.350069	0.3884835

Table 1: Estimated parameters for proposed MPS diode model. The thermal resistance (R<sub>th</sub>) was taken from data sheets.

**DMCS model presentation: diode C3D04060SS (dynamic and steady-state)**

\* (c) DMCS 2010-2011 M. Zubert, M. Napieralska, L. Starzak, G. Jablonski, M. Janicki, A. Napieralski

\*

.option list acct post=2 ingold=1 MEASDGT=7 ITRPRT=1

\*

**\* MODEL DEFINITION: A K Thermal**

**.SUBCKT DMCSC3D04060SS 100 200 300**

+ Rs=0.0253904 RsThC=7.788199E-5

+ RTh1=0.271 CTh1=7.07E-04

+ RTh2=1.69 CTh2=1.64E-03

+ RTh3=0.794 CTh3=0.220

+ CTh4=13.4

+ Alpha2=0.0 Alpha1=-4.127549E-05 Alpha0=0.01778526

+ Beta1=1.521115E-12 Beta2=0.03894504

+ Vintrsc1=0.948265 Vintrsc2=-1.042935E-3

+ Vref=0.0273944 AlphaRef1=2.70945E-3 AlphaRef2=0.0

+ VdLeft=100.0 VdRight=0.01

+ CJ0=251E-12 MJ=0.375303 VJ=0.3884835069962627

**\*\*\* Dynamic model**

EVDMINUS2 511 0 VALUE='(V(100,110)<=VJ) ? V(110,100):(-VJ) '

cqj 100 110 c='(time>0)? ((cj0\*pwr(v(511)+vj,-mj)) : 1.0e-12'

**\*\*\*Forward:**

EVDPlus 500 0 value='( V(100,110)>=0 )? V(100,110):0.0'

EIFwd 600 0 value='exp((V(500)-(Vintrsc1+ Vintrsc2\*V(300)))/(Vref\*(1+AlphaRef1\*(V(300)-300.15)+AlphaRef2\*(V(300)-300.15)\*(V(300)-300.15))))'

\* Interpolation: [0,VdRight]

EIFwdInterp 601 0 VALUE='V(500)\*exp((VdRight-(Vintrsc1+ Vintrsc2\*V(300)))/(Vref\*(1+AlphaRef1\*(V(300)-

+ 300.15)+AlphaRef2\*(V(300)-300.15)\*(V(300)-300.15))))'

\*

\*

```

***Reverse:
EVDMinus 510 0 VALUE='( V(100,110)<0,)? V(110,100):0.0'
EBetaT 810 0 VALUE='Beta1*exp(Beta2*V(310)) '
EAlphaT 820 0 VALUE=' Alpha2*V(310)*V(310)+Alpha1*V(310)+Alpha0 '
EIRev 800 0 VALUE=' V(810)*exp(V(820)*V(510)) '
* Interpolation: [VdLeft,0]
EIRevInterp 801 0 VALUE='(Exp(V(820)*VdLeft)*V(510)*V(810)*(VdLeft*(2-V(820)*VdLeft) +
+ V(510)*(V(820)*VdLeft-1))/(VdLeft*VdLeft)'
*
***All:
*Simplest formt: GIFwdRev 100 110 VALUE={ IF( V(100,110)>=0, V(600), -V(800)) }
GIFwdRev 100 110 VALUE='( V(100,110)>=0 )? ((V(500)>VdRight)?V(600):V(601)) :
+ ((V(100,110)<VdLeft)?(-V(800)):(-V(801))) '
RRs1 120 110 '0.5*Rs'
VIProbe 120 121 0
ERsT 121 130 VALUE=' RsThC*V(310)*(V(120,121)) '
RRs2 130 200 '0.5*Rs'
*
*
***Thermal domain:
VK2C 300 310 273.15
RTh3 350 310 'RTh3'
RTh2 360 350 'RTh2'
RTh1 370 360 'RTh1'
CTh4 310 0 'CTh4' IC=9.9
CTh3 350 0 'CTh3' IC='0.9+9.0*(Rth2+Rth1)/(Rth3+Rth2+Rth1)'
CTh2 360 0 'CTh2' IC='0.9+9.0*(Rth1)/(Rth3+Rth2+Rth1)'
CTh1 370 0 'CTh1' IC=0.9
* Can be used for steady-state and dynamic:
GTh1 0 370 VALUE='ABS(V(100,200)*I(VIProbe))'
.ENDS
*
*
* APPLICATION:
VIN 10 0 1.3 PULSE(1.3V -300V 50ns 25ns 25ns 50ns 300ns)
VthTemp 20 0 300.0
Rth5 21 20 0.588780
Cth5 21 0 106.849460
Rth4 22 21 0.353
XD1 10 11 22 DMCSC3D04060SS
*L1 10 11 100u
*L2 11 12 2.2u
*R2 11 12 56
Ron 11 0 0.1 ; 0.75
*
*.DC VIN START=-850 STOP=0 STEP=-5
*.DC VIN START=0 STOP=2.0 STEP=0.01
.TRAN 1ns 300ns
.print dc v(10) V(10,11) PAR('-i(vin)') V(22)
.print tran v(10) V(10,11) PAR('-i(vin)') v(22)
.END

```

Table 2: HSPICE model listing (dynamic and steady-state).




Figure 2: Comparision of the measured SiC diode forward characteristics with the proposed model. Temperatures: 25°C (green), 75°C (blue), 125°C for C3D04060 (black). Y-axis:  $V_{fwd}$  [V]; X-axis:  $\ln(I_{fwd})$ ; The measurement deviation is presented using error bars.

implemented in HSPICE environment. The dynamic behavior of MPS has been presented in Fig 4 bas on estimated junction capacitance parameters (see  $cj0$ ,  $vj$ ,  $mj$  in Table 1) based on datasheets for the junction temperature 300 K. The final model verivication will be obtained in the nearest feature using measurements from a real inverter circuit.




Figure 3: Dynamic simulated diode turn-off (C3D04060).

### 3 SUMMARY

The novel electro-thermal dynamic HSPICE model of SiC merged PiN Schottky diodes (2-nd and 3-rd generation) has been presented. The main advantage of the proposed model is that it is given in a closed form, which allows its straightforward implementation in the behavioural extensions of modern SPICE simulators as-well-as a very good agreement of steady-state characteristic with the behaviour of real devices in contrast to the models provided by the device manufacturers.

### 4 ACKNOWLEDGMENT

This work was supported partly by the grant of Polish Ministry of Science and Higher Education 0312/R/T02/2008/04 and the Internal University Grant K25/DZST/1/2011.

### REFERENCES

- [1] Zubert M et all: Static Electro-Thermal Model of SiC Merged PiN Schottky Diodes. W: 10th International Seminar on Power Semiconductors ISPS'10, Prague, Czech Republic, 1-3 Sep 2010, wyd. Czech Technical University in Prague, ed. V.Benda, 2010, s.282, s.227-232.
- [2] Zubert M. et all: "Behavioural Electro-Thermal Modelling of SiC Merged PiN Schottky Diodes" in book "Scientific Computing in Electrical Engineering", Series: Mathematics in Industry, Subseries: The European Consortium for Mathematics in Industry Springer-Verlag 2010 (in printing) France, Sep 19-24, 2010